Separable First-Order Differential Equations

We first illustrate Maple's differential equation solving ability by looking at an example that gives an explicit solution, $\frac{dy}{dx} = \frac{y-1}{x-3}$. Notice how we enter the differential equation. The dependent variable y is never entered by itself, but as $y(x)$, a function of the independent variable.

```
> restart;
> deq:=diff(y(x),x)=(y(x)-1)/(x-3);
```

$$deq := \frac{d}{dx}\, y(x) = \frac{y(x)-1}{x-3}$$

Let's see if this differential equation is separable by using MAPLE's computational power to test a DE for separability using the command odeadvisor. To use the command, we must first open the DEtools package by using the with command.

```
> with(DEtools);
```

[*AreSimilar*, *Closure*, *DEnormal*, *DEplot*, *DEplot3d*, *DEplot_polygon*, *DFactor*, *DFactorLCLM*, *DFactorsols*, *Dchangevar*, *Desingularize*, *FunctionDecomposition*, *GCRD*, *Gosper*, *Heunsols*, *Homomorphisms*, *IVPsol*, *IsHyperexponential*, *LCLM*, *MeijerGsols*, *MultiplicativeDecomposition*, *ODEInvariants*, *PDEchangecoords*, *PolynomialNormalForm*, *RationalCanonicalForm*, *ReduceHyperexp*, *RiemannPsols*, *Xchange*, *Xcommutator*, *Xgauge*, *Zeilberger*, *abelsol*, *adjoint*, *autonomous*, *bernoullisol*, *buildsol*, *buildsym*, *canoni*, *caseplot*, *casesplit*, *checkrank*, *chinisol*, *clairautsol*, *constcoeffsols*, *convertAlg*, *convertsys*, *dalembertsol*, *dcoeffs*, *de2diffop*, *dfieldplot*, *diff_table*, *diffop2de*, *dperiodic_sols*, *dpolyform*, *dsubs*, *eigenring*, *endomorphism_charpoly*, *equinv*, *eta_k*, *eulersols*, *exactsol*, *expsols*, *exterior_power*, *firint*, *firtest*, *formal_sol*, *gen_exp*, *generate_ic*, *genhomosol*, *gensys*, *hamilton_eqs*, *hypergeomsols*, *hyperode*, *indicialeq*, *infgen*, *initialdata*, *integrate_sols*, *intfactor*, *invariants*, *kovacicsols*, *leftdivision*, *liesol*, *line_int*, *linearsol*, *matrixDE*, *matrix_riccati*, *maxdimsystems*, *moser_reduce*, *muchange*, *mult*, *mutest*, *newton_polygon*, *normalG2*, *ode_int_y*, *ode_y1*, *odeadvisor*, *odepde*, *parametricsol*, *particularsol*, *phaseportrait*, *poincare*, *polysols*, *power_equivalent*, *rational_equivalent*, *ratsols*, *redode*, *reduceOrder*, *reduce_order*, *regular_parts*, *regularsp*, *remove_RootOf*, *riccati_system*, *riccatisol*, *rifread*, *rifsimp*, *rightdivision*, *rtaylor*, *separablesol*, *singularities*, *solve_group*, *super_reduce*, *symgen*, *symmetric_power*, *symmetric_product*, *symtest*, *transinv*, *translate*, *untranslate*, *varparam*, *zoom*]

```
> odeadvisor(deq,[separable,exact,linear]);
```

[*_separable*]

The odeadvisor command checks to see if the given DE is any of the types listed within the brackets. Let's see what happens for a DE that is not separable.

```
> deq2:=diff(y(x),x)=1+x*y(x);
```

$$deq2 := \frac{d}{dx}\, y(x) = 1 + x\, y(x)$$

```
> odeadvisor(deq2,[separable]);
```

$$[NONE]$$

The usual Maple command for solving a differential equation is dsolve. To find out about dsolve, you can do the following, which brings up the appropriate page from Maple's on-line help system. Just close the help window when you are done with it or choose the window separable. mw from the Window menu.

```
> ?dsolve
```

We solve the differential equation and assign the result to a variable for easy reference.

```
> soln:=dsolve(deq,y(x));
```

$$soln := y(x) = 1 + (x - 3)\ \_C1$$

Maple finds an explicit solution, a solution where $y(x)$ is written as a function of x. The $\_C1$ is an arbitrary constant. In other words, the differential equation has infinitely many solutions, one for each possible value of $\_C1$. But we should check that these really are solutions. We use the built in Maple command odetest.

```
> odetest(soln,deq);
```

$$0$$

The 0 indicates that we really do have a solution. Now we want to graph the solution for various values of $\_C1$. We substitute seven different values of $\_C1$ into the right hand side (rhs)of the equation, and save them as an array. Notice the structure of the "for" loop. We could suppress the output for this statement by replacing all of the semicolons with colons.

```
> for n from -3 to 3 do
  _C1:=n;
  sol[n]:=rhs(soln);
  od;
```

$$\_C1 := -3$$
$$sol_{-3} := 10 - 3\,x$$
$$\_C1 := -2$$
$$sol_{-2} := 7 - 2\,x$$
$$\_C1 := -1$$
$$sol_{-1} := 4 - x$$
$$\_C1 := 0$$
$$sol_0 := 1$$
$$\_C1 := 1$$
$$sol_1 := -2 + x$$
$$\_C1 := 2$$
$$sol_2 := -5 + 2\,x$$
$$\_C1 := 3$$
$$sol_3 := -8 + 3\,x$$

We now plot the seven graphs, each plot having a different color.

```
> plot([sol[-3],sol[-2],sol[-1],sol[0],sol[1],sol[2],sol[3]],x=-10.
```

```
.10);
```



Each value of $\_C1$ above leads to one particular solution of our differential equation. For instance, $\_C1 = 3$ leads to the solution $y(x) = 3x - 8$. Particular solutions can also be specified by an **initial condition** such as y(0) = -8, which gives this same particular solution. We call a differential equation with an initial value an **initial value problem (IVP)**. Here is the procedure for solving the IVP

$$\frac{dy}{dx} = \frac{y-1}{x-3}, \quad y(0) = -8$$

in Maple.

```
> deq:=diff(y(x),x)=(y(x)-1)/(x-3);
```

$$deq := \frac{\mathrm{d}}{\mathrm{d}x} y(x) = \frac{y(x)-1}{x-3}$$

```
> IC:=y(0)=-8;
```

$$IC := y(0) = -8$$

```
> dsolve({deq,IC},y(x));
```

$$y(x) = -8 + 3x$$

We see that we get the expected solution.

Some differential equations cannot be explicitly solved by Maple in a form where we can easily interpret the solution. Consider the equation $\frac{dy}{dx} = \frac{6x^5 - 2x + 1}{\cos(y) + e^y}$.

```
> restart;
> deq:=diff(y(x),x)=(6*x^5-2*x+1)/(cos(y(x))+exp(y(x)));
```

$$deq := \frac{d}{dx} y(x) = \frac{6x^5 - 2x + 1}{\cos(y(x)) + e^{y(x)}}$$

```
> dsolve(deq,y(x),explicit);
```

$$y(x) = RootOf\left(x^6 - x^2 + x - \sin(\_Z) - e^{\_Z} + \_C1\right)$$

Maple has found some sort of solution it can do some further computations with, but it is not very helpful to us. This expression indicates that our equation has multiple solutions for y in terms of x, and they are roots of the given expression. However, Maple can find an implicit solution.

```
> dsolve(deq,y(x),implicit);
```

$$x^6 - x^2 + x - \sin(y(x)) - e^{y(x)} + \_C1 = 0$$

Let's explore the development of this implicit solution further. We first rewrite the differential equation with variables separated.

```
> sepdeq:=diff(y(x),x)*(cos(y(x))+exp(y(x)))=6*x^5-2*x+1;
```

$$sepdeq := \left(\frac{d}{dx} y(x)\right)\left(\cos(y(x)) + e^{y(x)}\right) = 6x^5 - 2x + 1$$

We use the map command to integrate both sides of the equation with respect to x.

```
> intsepdeq:=map(Int,sepdeq,x);
```

$$intsepdeq := \int \left(\frac{d}{dx} y(x)\right)\left(\cos(y(x)) + e^{y(x)}\right) dx = \int \left(6x^5 - 2x + 1\right) dx$$

We get our implicit solution by using the value command to evaluate the indefinite integrals and add the constant of integration.

```
> solnsepdeq:=value(intsepdeq)+(0=C);
```

$$solnsepdeq := \sin(y(x)) + e^{y(x)} = x^6 - x^2 + C + x$$

Let's see what happens if we try to solve this implicit equation for an explicit solution.

```
> solve(solnsepdeq,y(x));
```

$$RootOf\left(\_Z - \ln\left(x^6 - x^2 + C + x - \sin(\_Z)\right)\right)$$

Now let's drop the implicit argument.

```
> soln:=dsolve(deq,y(x));
```

$$soln := x^6 - x^2 + x - \sin(y(x)) - e^{y(x)} + \_C1 = 0$$

So it looks as though if Maple cannot give a clean explicit solution, it defaults to an implicit solution. We again wish to check our solution.

```
> odetest(soln,deq);
```

$$0$$

This next statement loads the package plots, allowing for extra graphing capabilities.

```
> with(plots);
```

[*animate*, *animate3d*, *animatecurve*, *arrow*, *changecoords*, *complexplot*, *complexplot3d*, *conformal*, *conformal3d*, *contourplot*, *contourplot3d*, *coordplot*, *coordplot3d*, *densityplot*, *display*, *dualaxisplot*, *fieldplot*, *fieldplot3d*, *gradplot*, *gradplot3d*, *implicitplot*, *implicitplot3d*, *inequal*, *interactive*, *interactiveparams*, *intersectplot*, *listcontplot*, *listcontplot3d*, *listdensityplot*, *listplot*, *listplot3d*, *loglogplot*, *logplot*, *matrixplot*, *multiple*, *odeplot*, *pareto*, *plotcompare*, *pointplot*, *pointplot3d*, *polarplot*, *polygonplot*, *polygonplot3d*, *polyhedra_supported*, *polyhedraplot*, *rootlocus*, *semilogplot*, *setcolors*, *setoptions*, *setoptions3d*, *shadebetween*, *spacecurve*, *sparsematrixplot*, *surfdata*, *textplot*, *textplot3d*, *tubeplot*]

We want to graph this implicit solution with $\_C1=0$, assuming $y = f(x)$, using implicitplot. We first set the constant $\_C1=0$.

```
> onesoln:=subs(_C1=0,soln);
```

$$onesoln := x^6 - x^2 + x - \sin(y(x)) - e^{y(x)} = 0$$

So that **implicitplot** works correctly on all platforms, we need to replace each occurrence of $y(x)$ with y.

```
> onesoln2:=subs(y(x)=y,onesoln);
```

$$onesoln2 := x^6 - x^2 + x - \sin(y) - e^y = 0$$

We now do the implicit plot.

```
> implicitplot(onesoln2,x=-10..10,y=-30..30);
```



In drawing a graph, Maple evaluates the function at a predetermined number of points (the default being 50), and then joins the points by lines. Suspecting that this is not an accurate graph, we increase the numbers of points.

```
> implicitplot(onesoln2,x=-10..10,y=-30..30,numpoints=1000);
```



This looks better, but we try further.

```
> implicitplot(onesoln2,x=-10..10,y=-30..30,numpoints=10000);
```



We now view the graph of the equation in two other windows.

```
> implicitplot(onesoln2,x=-35..35,y=-35..35,numpoints=1000);
```




```
> implicitplot(onesoln2,x=20..30,y=20..30);
```


20.4
20.3
y
20.2
20.1
20
28.2 28.4 28.6 28.8 29.0 29.2 29.4 29.6 29.8 30.0
x
Curve 1